IBM Software

REXX Language Coding Techniques

Virgil Hein, IBM
vhein@us.ibm.com

May 2014

Important REXX Compiler Disclaimer

- **The information contained in this presentation is provided for informational purposes only.**
- **While efforts were made to verify the completeness and accuracy of the information contained in this presentation, it is provided "as is", without warranty of any kind, express or implied.**
- **In addition, this information is based on IBM's current product plans and strategy, which are subject to change by IBM without notice.**
- **IBM shall not be responsible for any damages arising out of the use of, or otherwise related to, this presentation or any other documentation.**
- **Nothing contained in this presentation is intended to, or shall have the effect of:**
  - Creating any warranty or representation from IBM (or its affiliates or its or their suppliers and/or licensors); or
  - Altering the terms and conditions of the applicable license agreement governing the use of IBM software.

Agenda

- **REXX compiler**
- **External environments and interfaces**
- **Key functions and instructions – power tools**
- **REXX data stack vs. compound variables**
- **EXECIO and stream I/O**
- **Troubleshooting**
- **Programming style and techniques**

The REXX Products

- **IBM Compiler for REXX on zSeries Release 4**
  - z/VM, z/OS: product number 5695-013
- **IBM Library for REXX on zSeries Release 4**
  - z/VM, z/OS: product number 5695-014
- **VSE part of operating system**
- **IBM Alternate Library for REXX on zSeries Release 4**
  - Included in z/OS base operating system (V1.9 and later)
  - Free download for z/VM (and z/OS)
    http://www.ibm.com/software/awdtools/rexx/rexxzseries/altlibrary.html

REXX Compiler Libraries

- **A REXX library is required to execute compiled programs**
  - Compiled REXX is not an LE language
- **Two REXX library choices:**
  - (Runtime) Library – a priced IBM product
  - Alternate library – a free IBM download
    - Uses the native system's REXX interpreter
- **At execution, compiled REXX will use whichever library is available:**
  - (Runtime) Library
  - Alternate Library

Why Use a REXX Compiler?

- **Program performance**
  - Known value propagation
  - Assign constants at compile time
  - Common sub-expression elimination
  - stem.i processing
- **Source code protection**
  - Source code not in deliverables
- **Improved productivity and quality**
  - Syntax checks all code statements
  - Source and cross reference listings
- **Compiler control directives**
  - %include, %page, %copyright, %stub, %sysdate, %systime, %testhalt

IBM Software

REXX External Environments

External Environments

- **ADDRESS instruction is used to define the external environment to receive host commands**
  - For example, to set TSO/E as the environment to receive commands

```
ADDRESS TSO
```

- **Several host command environments available in z/OS**
- **Other host command environments available in z/VM**

Host Command Environments in z/OS

- TSO
  - Used to run TSO/E commands like ALLOCATE and TRANSMIT
  - Only available to REXX running in a TSO/E address space
  - The default environment in a TSO/E address space
  - *TSO/E REXX Reference (SA22-7790)*
  - Example:

```
Address TSO "ALLOC FI(INDD) DA('USERID.SOURCE') SHR"
```

- MVS
  - Use to run a subset of TSO/E commands like EXECIO and MAKEBUF
  - The default environment in a non-TSO/E address space
  - *TSO/E REXX Reference (SA22-7790)*
  - Example:

```
Address MVS "EXECIO * DISKR MYINDD (FINIS STEM MYVAR"
```

Host Command Environments in z/OS

- ISPEXEC
  - Used to invoke ISPF services like DISPLAY and SELECT
  - Only available to REXX running in ISPF
  - *ISPF Services Guide (SC19-3626, SC34-4819)*
  - Example:

```
Address ISPEXEC "DISPLAY PANEL(APANEL)"
```

- ISREDIT
  - Used to invoke ISPF edit macro commands like FIND and DELETE
  - Only available to REXX running in an ISPF edit session
  - *ISPF Edit and Edit Macros (SC19-3621, SC28-1312)*
  - Example:

```
Address ISREDIT "DELETE .ZFIRST .ZLAST"
```

Host Command Environments in z/OS …

- CONSOLE
- LINK, LINKMVS, LINKPGM, ATTACH, ATTCHMVS, ATTCHPGM
- SYSCALL
- SDSF
- DSNREXX

Host Command Environments in z/OS …

- CONSOLE
  - Used to invoke MVS system and subsystem commands
  - Only available to REXX running in a TSO/E address space
  - Requires an extended MCS console session
  - Requires CONSOLE command authority
  - *TSO/E REXX Reference (SA22-7790)*
  - Example:

```
"CONSOLE ACTIVATE"
Address CONSOLE "D A"  /* Display system activity */
"CONSOLE DEACTIVATE"
```

Result:

```
IEE114I 04.50.01 2011.173 ACTIVITY 602
 JOBS     M/S    TS USERS    SYSAS    INITS   ACTIVE/MAX VTAM
  OAS
00002    00014    00002      00032    00005    00001/00020
  00010
```

Host Command Environments in z/OS …

- **LINK, LINKMVS, LINKPGM, ATTACH, ATTCHMVS, ATTCHPGM**
  - Host command environments for linking to and attaching unauthorized programs
  - Available to REXX running in any address space
  - LINK & ATTACH – can pass one character string to program
  - LINKMVS & ATTCHMVS – pass multiple parameters; half-word length field precedes each parameter value
  - LINKPGM & ATTCHPGM – pass multiple parameters; no half-word length field
  - *TSO/E REXX Reference (SA22-7790)*
  - Example:

```
"FREE FI(SYSOUT SORTIN SORTOUT SYSIN)"
"ALLOC FI(SYSOUT)   DA(*)"
"ALLOC FI(SORTIN)   DA('VHEIN.SORTIN') REUSE"
"ALLOC FI(SORTOUT)  DA('VHEIN.SORTOUT') REUSE"
"ALLOC FI(SYSIN)    DA('VHEIN.SORT.STMTS') SHR REUSE"
sortparm = "EQUALS"
Address LINKMVS "SORT sortparm"
```

Host Command Environments in z/OS …

- SYSCALL
  - Used to invoke interfaces to z/OS UNIX callable services
  - The default environment for REXX run from the z/OS UNIX file system
  - Use syscalls('ON') function to establish the SYSCALL host environment for a REXX run from TSO/E or MVS batch
  - *Using REXX and z/OS UNIX System Services (SA22-7806)*
  - Example:

```
call syscalls 'ON'
address syscall 'readdir / root.'
do i=1 to root.0
 say root.i
End
```

Result:

```
…
bin
dev
etc
…
```

Host Command Environments in z/OS …

- SDSF
  - Used to invoke interfaces to SDSF panels and panel actions
  - Use isfcalls('ON') function to establish the SDSF host environment
  - Use the ISFEXEC host command to access an SDSF panel
  - Panel fields returned in stem variables
  - Use the ISFACT host command to take an action or modify a job value
  - *SDSF Operation and Customization (SA22-7670)*

  - Example:

```
rc=isfcalls("ON")
Address SDSF "ISFEXEC ST"
do ix = 1 to JNAME.0
  if pos("VJHEIN",JNAME.ix) = 1 then do
    say "Cancelling job ID" JOBID.ix "for VJHEIN"
   Address SDSF "ISFACT ST TOKEN('"TOKEN.ix"') PARM(NP P)"
  end
end
rc=isfcalls("OFF")
exit
```

Host Command Environments in z/OS …

- DSNREXX
  - Provides access to DB2 application programming interfaces from REXX
  - Any SQL command can be executed from REXX
    - Only dynamic SQL supported from REXX
  - Use RXSUBCOM to make DSNREXX host environment available
  - Must CONNECT to required DB2 subsystem
  - Can call SQL Stored Procedures
  - *DB2 Application Programming and SQL Guide (SC19-4051)*

  - Example:

```
RXSUBCOM('ADD','DSNREXX','DSNREXX')
SubSys = 'DB2PRD'
Address DSNREXX "CONNECT" SubSys
Owner = 'PRODTBL'
RecordKey = 'ROW2DEL'
SQL_stmt = "DELETE * FROM" owner".MYTABLE" ,
           "WHERE TBLKEY = '"RecordKey"'"
Address DSNREXX "EXECSQL EXECUTE IMMEDIATE" SQL_stmt
Address DSNREXX "DISCONNECT"
```

Other External Environments in z/OS

- **IPCS**
  - Used to invoke IPCS subcommands from REXX
  - Only available when run from in an IPCS session
  - *MVS IPCS Commands (SA22-7594)*
- **CPICOMM, LU62, and APPCMVS**
  - Supports the writing of APPC/MVS transaction programs (TPs) in REXX
  - Programs can communicate using SAA common programming interface (CPI) communications calls and APPC/MVS calls
  - *TSO/E REXX Reference (SA22-7790)*

Other “Environments” and Interfaces in z/OS

- **System REXX**
  - A function package that allows REXX EXECs to be executed outside of conventional TSO/E and Batch environments
  - Can be invoked using assembler macro interface AXREXX or through an operator command
  - Easy way for Web Based Servers to run commands/functions and get back pertinent details
  - EXEC runs in problem state, key 8, in an APF authorized address space under the MASTER subsystem
  - Two modes of execution
    - TSO=NO runs in MVS host environment
address space shared with up to 64 other EXECs
limited data set support
    - TSO=YES runs isolated in a single address space
can safely allocate data sets
does not support all TSO functionality
  - *MVS Programming Authorized Assembler Services Guide (SA22-7605)*

Other “Environments” and Interfaces . . .

- **RACF Interfaces**
  - IRRXUTIL
    - REXX interface to R_admin callable service (IRRSEQ00) extract request
    - Stores output from extract request in a set of stem variables

```
myrc=IRRXUTIL("EXTRACT","FACILITY","BPX.DAEMON","RACF","","FALSE")
  say "Profile name: "||RACF.profile
  do a=1 to RACF.BASE.ACLCNT.REPEATCOUNT
    Say " "||RACF.BASE.ACLID.a||":"||RACF.BASE.ACLACS.a
  end
```

  - RACVAR function
    - Provides information from the ACEE about the running user
    - Arguments: USERID, GROUPID, SECLABEL, ACEESTAT

```
if racvar('ACEESTAT') <> 'NO ACEE' then
    say "You are connected to group " racvar('GROUPID')"."
```

  - *Security Server RACF Macros and Interfaces (SA22-7682)*

Other “Environments” and Interfaces . . .

- **Other ISPF Interfaces**
  - Panel REXX
    - Allows REXX to be run in a panel procedure
    - *REXX statement used to invoke it
    - REXX can be coded directly in the procedure or taken from a SYSEXEC or SYSPROC DD member
    - REXX can modify the values of ISPF variables
  - File Tailoring Skeleton REXX
    - Allows REXX to be run in a skeleton
    -)REXX control statement used to invoke it
    - REXX can be coded directly in the procedure or taken from a SYSEXEC or SYSPROC DD member
    - REXX can modify the values of ISPF variables
  - ISPF Dialog Developer’s Guide and Reference (SC19-3619, SC34-4821)

Host Command Environments in z/VM

- **CMS (default)**
  - Commands treated as if entered on the CMS command line
  - Same search order as CMS command line
- **COMMAND**
  - Basic CMS CMSCALL command resolution
    - To call an EXEC, prefix the command with the word EXEC
    - To send a command to CP, use the prefix CP
- **CPICOMM, CPIRR, OPENVM**

Key Instructions

Key Instructions – ARG, PULL, and PARSE

- **PARSE**
  - Allows the use of a template to split a source string into multiple components
  - Syntax:



- **ARG**
  - Retrieves the argument strings provided to a program or internal routine
    - Assigns them to variables
  - Short form for PARSE UPPER ARG
- **PULL**
  - Reads a string from the head of the external data queue
  - Short form for PARSE UPPER PULL
- **Good practice to use full commands vs short forms**

PARSE Templates

- **Simple template**
  - Divides the source string into blank-delimited words and assigns them to the variables named in the template

```
string = '  Parse the    blank-delimited   string'
parse var string var1 var2 var3 var4 .

var1 -> '  Parse'
var2 -> 'the'
var3 -> 'blank-delimited'
var4 -> 'string'
```

  - A period is a placeholder in a template
    - A "dummy" variable used to collect unwanted data

```
string = "Last one gets what's left"
parse var string var1 . var2

var1 -> "Last"
var2 -> "gets what's left"
```

    - Often used at the end of PARSE statement to take "the rest of the data"

PARSE Templates . . .

- **String pattern template**
  - A literal or variable string pattern indicating where the source string should be split

```
string = '  Parse the    blank-delimited   string'
```

Literal:

```
parse var string var1 '-' var2 .
```

Variable:

```
dlm = '-'
parse var string var1 (dlm) var2 .
```

Result (the same in both cases):

```
var1 -> '  Parse the    blank'
var2 -> 'delimited'
```

PARSE Templates . . .

- **Positional pattern template**
  - Use numeric values to identify the character positions at which to split data in the source string
  - An absolute positional pattern is a number or a number preceded by an equal sign

```
          ----+----1----+----2----+----3----+----4----+
string = 'Van Rite           Kevin          Australia  '
parse var string 1 surname 20 chrname 35 country 46 .

surname -> 'Van Rite            '
chrname -> 'Kevin          '
country -> 'Australia  '
```

  - A relative positional pattern is a number preceded by a plus or minus sign
    - Plus or minus indicates movement right or left, respectively, from the last match

```
          ----+----1----+----2----+----3----+----4----+
string = 'Van Rite           Kevin          Australia  '
parse var string 1 surname +19 chrname +15 country +11 .

surname -> 'Van Rite            '
chrname -> 'Kevin          '
country -> 'Australia  '
```

INTERPRET Instruction

- **Expression specified is evaluated**
  - Resulting value is processed (interpreted)
  - Adds an extra level of interpretation

```
conf = 'REXXLA'
interpret conf "= 'Memphis'; say 'Location is' REXXLA"
```

Result:

```
Location is Memphis
```

  - Provides powerful test and debugging capabilities

```
parse external debug_cmd     /* Receive command from user */
interpret debug_cmd          /* Run the user's command    */
```

  - Beware of security concerns

STORAGE Function

- **Syntax:**



- **Returns <length> bytes of data from the specified address in storage**
  - Address is a character string containing the hexadecimal representation of the storage address
  - Data is a character string that overwrites the data at address

```
data = storage(00FDE309,3)  /* Get 3 bytes at addr FDE309 */
```

- **A TSO/E external function but can be used in any MVS address space (TSO/E and non-TSO/E)**
- **Not all storage is available to access or update**
  - Virtual storage addresses may be fetch protected, update protected, or may not be valid
  - Null string returned

STORAGE Function . . .

- **To process addresses obtained with the STORAGE function**
  - C2D – character to decimal
    - Returns the decimal value of the binary representation of a string

```
C2D('81'X)      ->    129
```

  - D2X – decimal to hex
    - Returns a string, in character format, that represents a decimal number converted to hexadecimal

```
D2X(249)        ->    'F9'
```

  - Example – get the Address Space Vector Table address (CVTASVT) from the Communications Vector Table (CVT)

```
cvt = STORAGE(10,4)                          /* Get CVT address */
cvtasvt  = STORAGE(D2X(C2D(cvt)+556),4)      /* Get CVTASVT     */
```

STORAGE Function . . .

- **To simplify the job of retrieving pointers and other data**
  - PTR() - returns a 4 byte pointer as a decimal value
  - STG() - returns an EBCDIC string
    - First argument is the decimal value of the address where the data is located
    - Second argument is the length of the data to be returned
  - Example – get the MVS release and FMID from the CVT prefix area

```
NUMERIC DIGITS 20                /* Set precision to 20 digits   */
cvt = PTR(16)                    /* Get CVT address              */
cvtfixa  = cvt-256               /* CVT prefix address           */
cvtprod  = STG(cvtfixa+216,16) /* MVS product level data         */
Say 'MVS release and FMID:' cvtprod
PTR: RETURN C2D(STORAGE(D2X(arg(1)),4))     /* Return pointer */
STG: RETURN STORAGE(D2X(Arg(1)),Arg(2))     /* Return storage  */
```

  - Result:

```
MVS release and FMID: SP7.1.0 HBB7750
```

IBM Software

Data Stack and Compound Variables

What is a Data Stack?

- **An expandable data structure used to temporarily hold data items (elements) until needed**
- **When an element is needed it is always removed from the top of the stack**
- **A new element can be added either to the top (LIFO) or the bottom (FIFO) of the stack**
  - FIFO stack is often called a queue





Manipulating the Data Stack

- **3 basic REXX instructions**
  - PUSH - put one element on the top of the stack

```
elem1 = 'new top element'
PUSH elem
```

  - QUEUE - put one element on the bottom of the stack

```
elem2 = 'new bottom element'
QUEUE elem
```

  - PARSE PULL - remove an element from the (top) of the stack

```
PARSE PULL elem3
```

  - Result:

```
elem3 → 'new top element'
```

- **1 REXX function**
  - QUEUED() - returns the number of elements in the stack

```
num_elems = QUEUED()
```

Why Use the Data Stack?

- **To store a large number of data items of virtually unlimited size for later use**
- **Pass a large or unknown number of arguments between EXECs or routines**
- **Specify commands to be run when the EXEC ends**
  - Elements left on the data stack when an EXEC ends are treated as commands

```
Queue "TSOLIB RESET QUIET"
Queue "ALLOC FI(ISPLLIB) DA('ISP.SISPLOAD' 'SYS1.DFQLLIB') SHR REUSE"
Queue "TSOLIB ACTIVATE FILE(ISPLLIB) QUIET"
Queue "ISPF"
```

- **Pass responses to an interactive command that runs when the EXEC ends**

```
dest = SYSVAR('SYSNODE')"."USERID()
message = "Lunch time"
Queue "TRANSMIT"
Queue dest "LINE"
Queue message
Queue " "
```

Using Buffers in the Data Stack

- **An EXEC can create a buffer in a data stack using the MAKEBUF command**
- **All elements added after a MAKEBUF command are placed in the new buffer**
  - MAKEBUF basically changes where the QUEUE instruction inserts new elements
    - Remember QUEUE inserts at the “bottom” of the stack (or buffer)



Using Buffers in the Data Stack . . .

- **An EXEC can use MAKEBUF to create multiple buffers in the data stack**
  - MAKEBUF returns in the RC variable the number identifying the newly created buffer
- **DROPBUF command is used to remove a buffer from the data stack**
  - Allows an EXEC to easily remove temporary storage assigned to the data stack
  - A buffer number can be specified with DROPBUF to identify the buffer to remove
    - Default is to remove the most recently created buffer
  - DROPBUF 0 results in an empty data stack (use with caution)
- **The QBUF command is used to find out how many buffers have been created**
- **The QELEM command is used to find out the number of elements in the most recently created buffer**

- **Notes**
  - When an element is removed from an empty buffer, the buffer disappears.
  - To remove a buffer
    - Issue DROPBUF, or
    - Remove an element (via PARSE PULL) when the buffer is already empty
  - The next request to remove an element will move to the next buffer (if there is one)

Protecting Elements in the Data Stack

- **An EXEC can use the stack, but protect itself from inadvertently removing someone else's data stack elements**
  - Create a new private data stack using the NEWSTACK command
- **All elements added after a NEWSTACK command are placed in the new data stack**
  - Elements on the original data stack cannot be accessed by an EXEC or any called routines until the new stack is removed (not just emptied)
  - When there are no more elements in the new data stack, information is taken from the terminal (not the original data stack)



Protecting Elements in the Data Stack . . .

- **DELSTACK - removes a data stack**
  - Removes the most recently created data stack
    - Including all remaining elements in the stack
  - CAUTION
    - If no stack previously created with NEWSTACK, then DELSTACK removes all the elements from the original stack
- **QSTACK - returns the number of data stacks**
  - Including the original stack
  - Puts the value in the variable RC
- **NOTE: The QUEUED() function returns the number of elements in the current data stack**

What is a Compound Variable?

- **A series of symbols (simple variable or constant) separated by periods**
- **Made up of 2 parts – *stem* and *tail***
  - *stem* is the first symbol and the first period
    - Symbol must be a name
    - Sometimes called the *stem variable*
  - *tail* follows the stem and comprises one or more symbols separated by periods
    - Symbol is often a number, but not required to be
- Variables take on previously assigned values
  - **If no value assigned, takes on the uppercase value of the variable name**

```
day.1                                   stem:   DAY.
                                        tail:   1

array.i                                 stem:   ARRAY.
                                        tail:   I

name = 'Smith'
phone = 12345

employee.name.phone                     stem:   EMPLOYEE.
                                        tail:   Smith.12345
```

Compound Variable Values

- **Initializing a stem to some value automatically initializes every compound variable with the same stem to the same value**

```
say month.15          ⟶  MONTH.15
month. = 'Unknown'
month.6 = 'June'
month.3 = 'March'

say month.15          ⟶  Unknown
val = 3
say month.val         ⟶  March
```

- **Easy way to reset the values of compound variables**

```
month. = ''
say month.6           ⟶  ''
```

- **DROP instruction can be used to restore compound variables to their uninitialized state**

```
drop month.
say month.6           ⟶  MONTH.6
```

Processing Compound Variables

- **Compound variables provide the ability to process one-dimensional arrays**
  - Use a numeric value for the tail
  - Good practice to store the number of array entries in the compound variable with a tail of 0 (zero)
  - Often processed in a DO loop using the loop control variable as the tail

```
invitee.0 = 10
do i = 1 to invitee.0
  SAY 'Enter the name for invitee' i
  PARSE PULL invitee.i
end
```

- **Stems can be used with the EXECIO command to read data from and write data to a data set**
- **Stems can also be used with the OUTTRAP external function to capture output from commands**

Processing Compound Variables . . .

- **The tail for a compound variable can be used as an index to related data**
- **Given the following input data:**

```
Symbol   Atomic#    Name         Weight
H        1          Hydrogen     1.00794
HE       2          Helium       4.002602
LI       3          Lithium      6.941
...
```

- **The unique symbol value can be used as the tail of compound variables that hold the rest of the symbol's values**

```
"EXECIO * DISKR INDD (STEM rec. FINIS"
Do i = 2 To rec.0
  Parse Var rec.i symbol atomic.symbol name.symbol weight.symbol
End i
Say "Which atomic symbol do you want to learn about?"
Parse Pull symbol
Say "The name of" symbol "is" name.symbol"."
Say "The atomic number for" symbol "is" atomic.symbol"."
Say "The atomic weight of" symbol "is" weight.symbol"."
```

Data Stack vs Compound Variables

- **Data Stack**
  - Advantages
    - Can be used to pass data to external routines
    - Able to specify commands to be run when the EXEC ends
    - Can provide response(s) to an interactive command that runs when the EXEC ends
  - Disadvantages
    - Program logic required for stack management
    - Processing needs 2 steps
      - Take data from input source and store in stack
      - Read from stack into variables
    - Stack attributes and commands are OS dependent

Data Stack vs Compound Variables . . .

- **Compound Variables**
  - Advantages
    - Basically variables - REXX will manage them like other variables
    - Only one step required to assign a value
    - Provide opportunities for clever and imaginative processing
  - Disadvantages
    - Can not be used to pass data between external routines
- **Conclusion**
  - Try to use compound variables whenever appropriate
    - They are **simpler**

IBM Software

I/O and Troubleshooting

EXECIO Command

- **A TSO/E REXX command that provides record-based processing**
  - Used to read and write records from/to a sequential data set or partitioned data set member
  - Requires a DDNAME to be specified
    - Use ALLOC command to allocate data set or member to a DD
- **Records can be read into or written from compound variables or the data stack**
- **Can also be used to:**
  - Open a data set without reading or writing any records
  - Empty a data set
  - Copy records from one data set to another
  - Add records to the end of a sequential data set
  - Update data in a data set, one record at a time

REXX Stream I/O

- **Function package shipped with z/OS**
  - Also shipped with the IBM Library for Rexx on zSeries
- **Allows REXX EXECs to use stream I/O functions to process sequential data sets and partitioned data set members**
- **Why use stream I/O?**
  - Extends and enhances I/O capabilities of REXX for TSO/E
  - Shields the complexity of z/OS data set I/O
    - (To some degree)
  - A familiar I/O concept
  - Provides better portability of REXX between OS platforms

Troubleshooting – Condition Trapping

- **CALL ON and SIGNAL ON instructions can be used to trap exception conditions**
- **Syntax:**



- **Condition types:**
  - ERROR - error upon return (positive return code)
  - FAILURE - failure upon return (negative return code)
  - HALT - an external attempt was made to interrupt and end execution
  - NOVALUE - attempt was made to use an uninitialized variable
  - SYNTAX - language processing error found during execution

Troubleshooting – Condition Trapping. . .

- **Good practice to enable condition handling to process unexpected errors**
- **Use REXX provided functions and variables to identify and report on exceptions**
  - CONDITION function – returns information on the current condition
    - Name and description of the current condition
    - Indication of whether the condition was trapped by SIGNAL or CALL
    - Status of the current trapped condition
  - RC variable – return code
    - Contains the command return code for ERROR and FAILURE
    - Contains the syntax error number for SYNTAX
  - SIGL variable – line number of the clause that caused the condition
  - ERRORTEXT function – returns REXX error message for a SYNTAX condition

```
say ERRORTEXT(rc)
```

  - SOURCELINE function – returns a line of source from the REXX EXEC

```
say SOURCELINE(sigl)
```

Troubleshooting – Trace Facility

- **Provides powerful debugging capabilities**
  - Displays the results of expression evaluations
  - Displays the variable values
  - Follows the execution path
  - Interactively pauses execution and runs REXX statements
- **Activated using the TRACE instruction and function**
- **Syntax:**



Troubleshooting – Trace Facility . . .

- **Code example:**

```
A = 1
B = 2
C = 3
D = 4
Trace I
If (A > B) | (C < 2 * D) Then
  Say 'At least one expression was true.'
Else
  Say 'Neither expression was true.'
```

- **Result:**

```
 6 *-* If (A > B) | (C < 2 * D)
       >V>   "1"
       >V>   "2"
       >O>   "0"
       >V>   "3"
       >L>   "2"
       >V>   "4"
       >O>   "8"
       >O>   "1"
       >O>   "1"
       *-*  Then
     7 *-*  Say 'At least one expression was true.'
       >L>    "At least one expression was true."
At least one expression was true.
```

Troubleshooting – Trace Facility . . .

- **Interactive trace provides additional debugging power**
  - Pause execution at specified points
  - Insert instructions
  - Re-execute the previous instruction
  - Continue to the next traced instruction
  - Change or terminate interactive tracing
- **Starting interactive trace**
  - ? Option with the TRACE instruction
  - EXECUTIL TS command
    - Code in your REXX EXEC
    - Issue from the command line to debug next REXX EXEC run
  - Cause an attention interrupt and enter TS

Programming Style and Techniques

- **Be consistent with your style**
  - Helps others read and maintain your code
  - Having style rules will make the job of coding easier
- **Indentation**
  - Improves readability
  - Helps identify unbalanced or incomplete structures (DO-END pairs)
- **Comments**
  - Provide them!
  - Choices:
    - In blocks
    - To the right of the code
- **Capitalization**
  - Can improve readability
  - Suggestion - use all lowercase except
    - Labels
    - Calls to internal subroutines

Programming Style and Techniques . . .

- **Variable names**
  - Try to use meaningful names – helps understanding and readability
  - Avoid 1 character names – easy to type but difficult to manage and understand
- **Subroutines**
  - Try to avoid the over use of subroutines or functions
  - Subroutines are useful, but have performance impact
  - If it's only called once, does it need to be a subroutine?
- **Comparisons**
  - REXX supports *exact* (e.g. "==") and *inexact* (e.g. "=") operators
  - Only use *exact* operators when appropriate
    ```
    if a == "SAVE" then …
    ```
  - Above comparison will fail if **a** is **"SAVE "**
  - Avoid using the NOT ("¬") character
    - Portability problem when transferring code to an ASCII platform
    - Use "<>", "/=", or "\="

Programming Style and Techniques . . .

- **Semicolons**
  - Can be used to combine multiple statements in one line
    - DON'T – detracts from readability
  - Languages like C and PL/I require a ";" to terminate a line
  - Can also be done in REXX
    - DON'T – doubles internal logic statement count for interpreted REXX
- **Conditions**
  - For complex statements, REXX evaluates all Boolean expressions, even if first fails:

```
if 1 = 2 & 3 = 4 & 5 = 5 then say 'Impossible'
```

    - Divide-by-zero can still occur if a=0

```
if a \== 0 & b/a > 1 then ...
```

    - Can be avoided by nesting IF statements:

```
if a \== 0 then
  if b/a > 1 then ...
```

Programming Style and Techniques . . .

- **Literals**
  - Important to use literals where appropriate
    - For example: external commands
  - Lazy programming can lead to unfortunate results
    - For uninitialized variables: value=name
      ```
      control errors cancel
      ```
    - This usually works
      - Breaks if any of the 3 words is a variable with value already assigned
    - Also a performance cost for unnecessary variable lookups (20% + more CPU)
    - Instead enclose literals in quotation marks
      ```
      "control errors cancel"
      ```

Open Object REXX

- **Open Object REXX is available via open source community**
  - Runs on Linux on on System z
  - Many other platforms
- **www.oorexx.org**
- **90%+ compatible with other System z REXX programs**
- **Informal testing with SLES on memory and CPU constrained system**
  - **Compare PERL and OOREXX – OOREXX is much faster!**
  - **Memory footprint of OOREXX is similar to PERL with several modules loaded**

Additional Information and Contacts

- **REXX Compiler User's Guide and Reference http://publibfi.boulder.ibm.com/epubs/pdf/h1981605.pdf**
- **IBM REXX Website http://www.ibm.com/software/awdtools/rexx**
- **IBM Contacts**
  - Virgil Hein, vhein@us.ibm.com
    - Compiler and Library for REXX on zSeries
  - George Kochanowski, jjkoch@us.ibm.com
    - REXX Compiler

REXX Coding question:

- Platform: OS/2
  Question:
  * For just an entered command, I can re-direct the output from the command by using the ">" character.
    PING domain.com 56 100 > d:\SavePING.Txt

    Or, in a REXX Cmd, I can use:
    e.g.,
    ```
    FileOut = "d:\Dir1.SubDir2\SavePING.Txt"
    'PING domain.com 56 100 1>> ' || FileOut
    ```

  * Problem: I want to be able to document the info that is put into d:\SavePING.Txt with the "SAY" statement.
    e.g.,
    ```
    Say "Today's Date/Time: " || Date( ) || " / " || Time( );
    ```
    - I've tried using the '>' but that does not work.
    - I've tried using '1>>' that works for the PING as above.

    NOTHING I've tried works with 'Say'.

  HOW do I do this with REXX in OS/2? [Actually, not pure OS/2, but eComStation? See www.eComStation.Com for more info about the platform.

REXX Coding question, cont.:

- Carl almost has it correct.

The '>' operator always causes the output file to be truncated and then the output is added to the zero-length file.

The '>>' operator always just appends the output to the end of the file. If the file does not exist it creates it.

So use the '>' operator on the first command and then use the '>>' operator on all subsequent commands.

- Well, the description that I had was quite similar to what you provided. But, it does not work as described.

  Here is a quick test REXX Script, with several variations:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

/* CmdName: SayTest.Cmd                                    */
/*          Test redirection of Say statement              */
/*********************************************************/
Parse arg YorN
If YorN = "Y"
Then
Say "Current Date/Time:  " || DATE( ) || " / " TIME( ) >>
g:\CatchSay.Txt
Else
Say "Current Date/Time:  " || DATE( ) || " / " TIME( )

>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

  And, the results of executing the above:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

[O:\]saytest
Current Date/Time:  17 Mar 2014 /  15:48:42

[O:\]saytest Y
     7 +++     Say 'Current Date/Time:  ' || DATE() || ' / ' TIME() >>
g :
\ Cat
chSay.Txt;
REX0035: Error 35 running F:\CmdFiles\SayTest.Cmd, line 7: Invalid
expression

<<CHANGE THE REDIRECTION CODE from ">>" to ">"

[O:\]saytest Y
     7 +++     Say 'Current Date/Time:  ' || DATE() || ' / ' TIME() > g
.
\ CatchSay.Txt;
REX0035: Error 35 running F:\CmdFiles\SayTest.Cmd, line 7: Invalid
expression

[O:\]

>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

- Changing the target file to a variable instead of the constant name:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

/* CmdName: SayTest.Cmd                                    */
/*          Test redirection of Say statement              */
/*****************************************************************/
Catchit = "g:\CatchSay.Txt"
Parse arg YorN
If YorN = "Y"
Then
Say "Current Date/Time:  " || DATE( ) || " / " TIME( ) >> Catchit
Else
Say "Current Date/Time:  " || DATE( ) || " / " TIME( )

>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

Results in:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

[O:\]saytest Y
0

[O:\]

>
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

Using the filename as a constant string instead of a variable
changes nothing.

Putting a '1' in front of the '>>' with the original command
goes back to the 'invalid expression' message:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

[O:\]saytest Y
     8 +++      Say 'Current Date/Time:  ' || DATE() || ' / ' TIME() 1
>> g
: \ CatchSay.Txt;
REX0035: Error 35 running F:\CmdFiles\SayTest.Cmd, line 8: Invalid
expression

>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

- The SAY instruction can not be redirected with either the '>' or the
'>>'
operator. They are only recognized by the shell. So you will need to
substitute the SAY instruction with

'echo "Current Date/Time: "' || DATE( ) || '" / "' TIME( ) '"'>>
g:\CatchSay.Txt"'

Note where the single quotes are.

- That doesn't work either.

  I simply did a cut/paste of your example into the REXX Cmd.
  I also put a '@' in front of the ECHO to keep the SAY itself
  from displaying. The result:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
[O:\]saytest n
"Current Date/Time:  "18 Mar 2014" / " 12:12:26 ">>  g:\CatchSay.Txt"

Redirection attempt...

Current Date/Time:  18 Mar 2014 /  12:12:26

[O:\]

>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

  It just 'echos' the redirection syntax and nothing goes to the file.

  I had actually solved the problem with:

```
>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

FileOut="V:CatchSay.Txt"

/*************************************************************/

RC=LINEOUT(FileOut,"Current Date/Time:  " || DATE() || " / " || TIME())

>~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
```

  This is working, but I have to be careful to close the file
  before trying to redirect subsequent command output to the
  file.

धन्यवाद

Hindi

多謝

Traditional Chinese

감사합니다

Korean

Спасибо

Russian

Gracias

Spanish

Thank You

English

شكراً

Arabic

Obrigado

Brazilian Portuguese

Danke

German

Grazie

Italian

多谢

Simplified Chinese

Merci

French

நன்றி

Tamil

ありがとうございました

Japanese

ขอบคุณ

Thai